関数のスクリプト-その0（ノート）

SHIMURA Masato
jcd02773@nifty.ne.jp

2009年1月26日

目次

概要

関数の J 言語の Tacit 定義に関するレビューである。

1　関数定義入門

1.1　& @ hook fork

$f(x) = x^2 + 1$	`f1=:>:@*:`	`f1 >:i.5`
		
		`2 5 10 17 26`

$f(x) = \sqrt{x+5}$	`f2=:%:@(5:+])`	`(5&+)4 11 20`
		`9 16 25`
		
		`f2 4 11 20`
		`3 4 5`

$f(x) = 4x$	`f3=:*&4`	`f3 >: i.5` `4 8 12 16 20`

$f1, f2, f3$	`f4=:f1,f2,:f3`	`6j2&": f4 >:i.5`
		`2.00 5.00 10.00 17.00 26.00`
		`2.45 2.65 2.83 3.00 3.16`
		`4.00 8.00 12.00 16.00 20.00`

$f(x) = x^2 - y^2$	`f5=:-&*:`	`^&2] 3 4 5,: 1 2 3`
		`9 16 25`
		`1 4 9`
		
		`3 4 5 f5 1 2 3`
		`8 12 16`

$\sqrt{x+y}$	`f11=:%:@+`	`   1 3 6 f11 3 6 10` `2 3 4`

x^2
$(x+y)^2$
	`f12=: *:@+`	`  f12 3 4 5  NB. monad`
		`9 16 25`
		
		`1 2 3  f12 3 4 5 NB. dyad`
		`16 36 64`

$\sqrt{xy}$	`f13=: %:@*`	`1 2 3 f13 4 8 12` `2 4 6`
x^2+y^2	`f14=: +&*:`	`1 2 3 f14 4 8 12` `17 68 153`

$f(x)=(x-y)(x+y)$	`f15=:+ * -`	`   2 4,: 3 6`
		`2 4`
		`3 6`
		
		`   2 4 f15 3 6`
		`_5 _20`

$x(x-1)$	`f16=: [ *  <:`	`   3 6 ,: 2 5`
		`3 6`
		`2 5`
		`    f16 3 6`
		`6 30`

$x + x^2$	`f17=:] + *:`	`f17 3 6` `12 42`
$x\sqrt{y}$	`f18=:[ * %:@]`	`3 5  f18 4 9` `6 15`

1.2 &.

`x(f &. g)y` $\iff g^{-1}(g(x)\ f\ g(y))$

$(\sqrt{x} + \sqrt{y})^2$	`f6=: +&.%:`	`4 9 ,: 16 25`
		` 4  9`
		`16 25`
		
		`4 9 f6 16 25`
		`36 64`

$\sqrt{x^2 - y^2}$	`f7=: -&.*:`	`3 5 f7 2 4` `2.23607 3`

2 多項式 polinomial

$(x - x_1)(x - x_2)(x - x_3)(x - x_4)$
$= x^4$
$- x^3(x_1 + x_2 + x_3 + x_4)$
$+ x_2(x_1x_2 + x_1x_3 + x_1x_4 + x_2x_3 + x_2x_4 + x_3x_4)$
$- x(x_1x_2x_3 + x_1x_2x_4 + x_1x_3x_4 + x_2x_3x_4)$
$+ x_1x_2x_3x_4$

$(x - x_1)(x - x_2)(x - x_3)$
$= x^3$
$- x^2(x_1 + x_2 + x_3)$
$+ x(x_1x_2 + x_1x_3 + x_2x_3)$

$-x_1x_2x_3$

```
NB. polinomial
ppr=: +//.@(*/)
cfr=: [: ppr/ - ,. 1:
```

$p(x) = 2x^2 + 1$
$q(x) = x^2 - 4x - 3$
$p(x)q(x) = 2x^4 - 8x^3 - 5x^2 - 4x - 3$
	$p(x) = 2x^2 + 1$ NB. `1 0 2&p.`
	$q(x) = x^2 - 4x - 3$ NB. `_3 _4 1&p.`
	`   1 0 2 ppr _3 _4 1`
	`_3 _4 _5 _8 2`

$(x-5)(x+2)(x-3) = 30 - x - 6x^2 + x^3$	`   cfr 5 _2 3`
	`30 _1 _6 1`
	NB. 入力符号が逆

3 差分と微分

3.0.1 差分

差分 (D :) は Δ の大きさを指定する。

```
cube=: ^&3 "0
;("1),.({@> 0.1 0.01 0.001 0.0001)  cube D:(1)(L:0) 1 2 3 4 5

  1       2       3       4       5
  -------------------------------------
3.31   12.61   27.91   49.21   76.51    NB. 0.1
3.0301 12.0601 27.0901 48.1201 75.1501  NB. 0.01
3.003  12.006  27.009  48.012  75.015   NB. 0.001
3.0003 12.0006 27.0009 48.0012 75.0015  NB. 0.0001
```

3.0.2 微分

```
   cube d.1 ] 1 2 3 4 5
3 12 27 48 75
```

微分の階数は d. の引数で与える。

```
 |:  cube d. (i.4) 1 2 3 4 5

1   8 27 64 125    NB. x^3
3 12 27 48  75    NB. 3x^2
6 12 18 24  30    NB. 6x
6  6  6  6   6    NB. 6
```

cube =: x^3 は多項式でも定義できる。

```
 |:  0 0 0 1&p. d.(i.4) 1 2 3 4 5
1   8 27 64 125
3 12 27 48  75
6 12 18 24  30
6  6  6  6   6
```

```
    cube d.1 ] 1 2 3 4 5
 3 12 27 48 75
```

```
    cube D.(1) 1 2 3 4 5
3 12 27 48 75

 0 0 0 1&p. D.(1) 1 2 3 4 5
3  0  0  0  0
0 12  0  0  0
0  0 27  0  0
0  0  0 48  0
0  0  0  0 75
```

3.0.3　Example

$f(x) = x^5 - x^3 + 1$
$\dot{f}(x) = 5x^4 - 3x^2$
$\ddot{f}(x) = 20x^3 - 6x$
$\dddot{f}(x) = 60x^2 - 6$

`1 0 0 _1 0 1&p.` $f(x) = x^5 - x^3 + 1$

```
   1 0 0 _1 0 1&p. (D.1) 3  NB. x=3
378
```

$\dot{f}(x) = 5x^4 - 3x^2$

```
   1 0 0 _1 0 1&p. (D.2) 3
522.644
```

```
   1 0 0 _1 0 1&p. (d.2) 3
522
```

```
   1 0 0 _1 0 1&p. (D.1)(D.1) 3
522
```

4 テーラー展開

4.1 power series

power series expantion ベキ級数展開

$y = a_0 + a_1 x + a_2 x^2 + \cdots = \sum_{n=0}^{\infty} a_n x^n$

次のようなものがイメージできる。

1. 折り畳めるへら鮒釣りの継ぎ竿
2. ロシアの入れこ人形マトリョーシカ
3. 小編成の貨物列車

4.2 テーラー展開

4.2.1 e

!	0! 1! 2! 3! 4!	`! i.5` `1 1 2 6 24`

e	$e \approx 1 + 1 + \frac{1}{2!} + \frac{1}{3!} + \frac{1}{4!} + \cdots = \sum_{n=0}^{\infty} \frac{1}{n!}$	`(% & !) i.5`
		`1 1 0.5 0.166667 0.0416667`
		
		`^t.i.5`
		`1 1 0.5 0.166667 0.0416667`
		
		`+/ (% & !) i.10`
		`2.71828`

e^x	$e^x \approx 1 + \frac{x}{1!} + \frac{x^2}{2!} + \frac{x^3}{3!} + \frac{x^4}{4!} + \cdots + \frac{x^n}{n!}$	`6 (([^i.@[),: !&i.@[) 5`
		`1 5 25 125 625 3125`
		`1 1  2   6  24  120`
		
		`ex=: +/@(([^i.@[) %/"0 !&i.@[)`
		`   15 ex 5`
		`148.38`
		
		NB.　e^5 **を** 15 **項まで展開する**
		(x **=** 15**）項程度で収束する。**
		
		`   ^5`
		`148.413`
		`   1x1^5`
		`148.413`

4.3　sin, cos

$$e^x \approx 1 + \frac{x}{1!} + \frac{x^2}{2!} + \frac{x^3}{3!} + \frac{x^4}{4!} + \cdots + \frac{x^n}{n!}$$
$$sinx = x - \frac{x^3}{3!} + \frac{x^5}{5!} - \frac{x^7}{7!}$$
$$cosx = 1 - \frac{x^2}{2!} + \frac{x^4}{4!} - \frac{x^6}{6!}$$

```
   6 ex0 0.5
1 0.5 0.25 0.125 0.0625 0.03125
1   1    2     6     24      120

  %/ 6 ex0 0.5
1 0.5 0.125 0.0208333 0.00260417 0.000260417
```

sin,cos のテーラー展開の係数の符号は後出の weighted taylor(t:) に任そう。円関数は収束が早いので 6 項までとする。

```
   (1&o.) t: i.6
0 1 0 _1 0 1
```

```
   (2&o.) t: i.6
1 0 _1 0 1 0
```

$$sinx = x - \frac{x^3}{3!} + \frac{x^5}{5!} - \frac{x^7}{7!}$$

```
 1&o. w_taylor 0.5 8 NB. 8 項まで
+--------+--------+-----------------------------------------+
|0.479426|0.479426|0 0.5 0 _0.0208333 0 0.000260417 0 _1.5501e_6|
+--------+--------+-----------------------------------------+
 sin 0.5  sum taylor  taylor
```

$$cosx = 1 - \frac{x^2}{2!} + \frac{x^4}{4!} - \frac{x^6}{6!}$$

```
  2&o. w_taylor 0.5 8
+--------+--------+-------------------------------------+
|0.877583|0.877582|1 0 _0.125 0 0.00260417 0 _2.17014e_5 0|
+--------+--------+-------------------------------------+
 cos 0.5  sum taylor taylor
```

```
 ex0=:(([ ^ i.@[) ,: !&i.@[)

w_taylor=: 1 : 0
NB. Usage: 1&o. u 0.5 6
NB. value 0.5 / 6 jisuu
if. 1=# y do. X0=. 6
else. 'X0 Y0'=. |. y end.
f=: u
TMP0=:f t: i. X0
TMP1=:X0 ex0  Y0
(u {.y);(+/ TMP2);TMP2=:TMP0 *"1 %/ TMP1
)
```

5　J のテーラー関数

J のテーラー関数として `t.` `T.` `t:`の 3 種類が用意されている。

5.1　t.(taylor Coefficient)T. (taylor approximation)

5.1.1　e, sin, cos

J の `t.` は無蓋の小編成の貨物列車のようで、台車だけが用意されており、積み荷は自分で積むこととなる。

e	`   ^t.  i.6  NB. 6` **項まで**
	`1 1 0.5 0.166667 0.0416667 0.00833333`
	
	`    +/ ^t. i.6`
	`2.71667`
	
	`   ^T.6] 1 NB.6` **項まで**
	`2.71667`
	
	`   ^T._ ]1`
	`2.71828`

sin	`   1&o. t.  i.6`
	`0 1 0 _0.166667 0 0.00833333`
	
	`+/ (0.5^i.6)*  1&o. t.  i.6`
	`0.479427`
	
	`   1&o. T. 6] 0.5`
	`0.479427`
	
	`   1&o. 0.5`
	`0.479426`

T. は項数を指定する。

cos	`   2&o. t.  i.6`
	`1 0 _0.5 0 0.0416667 0`
	
	` (0.5^i.6)*  2&o. t.  i.6`
	`1 0 _0.125 0 0.00260417 0`
	
	`+/ (0.5^i.6)*  2&o. t.  i.6`
	`0.877604`
	
	`  +/ 2&o. T. 6]0.5`
	`0.877604`
	
	`   2&o. 0.5`
	`0.877583`

5.2 Weighted taylor

$$
\begin{aligned}
f(x) &= e^x \quad so \quad f(0) = 1\\
f'(x) &= e^x \quad so \quad f'(0) = 1\\
f''(x) &= e^x \quad so \quad f''(0) = 1\\
f^3(x) &= e^x \quad so \quad f^3(0) = 1\\
f^4(x) &= e^x \quad so \quad f^4(0) = 1\\
f^5(x) &= e^x \quad so \quad f^5(0) = 1
\end{aligned}
$$

formula for the 5th degree approximation to $f(x)$ is

$$f(x) = f(0) + f'(0) + \frac{f''(0)x^2}{2!} + \frac{f^3(0)x^3}{3!} + \frac{f^4(0)x^4}{4!} + \frac{f^5(0)x^5}{5!}$$

$$
\begin{aligned}
e^x &= 1 + x + \frac{x^2}{2!} + \frac{x^3}{3!} + \frac{x^4}{4!} + \frac{x^5}{5!}\\
e^1 &= 1 + 1 + \frac{1}{2!} + \frac{1}{3!} + \frac{1}{4!} + \frac{1}{5!}
\end{aligned}
$$

```
   ^ t: i.6
1 1 1 1 1 1
```

```
   sin t: i.6
0 1 0 _1 0 1

   cos t: i.6
1 0 _1 0 1 0
```

5.3 テーラー関数の応用

5.3.1 polynomial

$1+2x+x^2$ $1+3x+3x^2+x^3$	`f=: 1 2 1 &p.` `g=: 1 3 3 1 &p.`

$1+2x+x^2$
$1+3x+3x^2+x^3$
×———————-
$1+5x+10x^2+10x^3+5x^4+x^5$
	`(f*g) t. i.8`
	
	`1 5 10 10 5 1 0 0`

	`(f i.8),.(g i.8),.`
	`((f*g) t. i.8)&p. i.8`
	
	`f    g    f*g`
	` 1    1      1`
	` 4    8     32`
	` 9   27    243`
	`16   64   1024`
	`25  125   3125`
	`36  216   7776`
	`49  343  16807`
	`64  512  32768`

5.3.2

t. はどのように使うか、どこまで汎用性があるか

複素数	`6&o.@j. t. i.6 NB. cosh` `1 0 _0.5j6.12323e_17 0 0.0416667j_1.02054e_17 0`
$\frac{1}{e^x}$	`%@^ t. i.6` `1 _1 0.5 _0.166667 0.0416667 _0.00833333`

lnx	`^. t. i.6`
	`|domain error`
	`|       ^.t.i.6`

$\sqrt{x}$	`%: t. i.6`
	`|domain error`
	`|       %:t.i.6`

付録 A　接続詞　ボンド、コンポーズ (&) とアトップ (@)

接続詞は動詞を連結して複数の作用を一度に行う。（複合動詞　持って走る）接続詞には右引数を一個取る単項型と *x u*&*v y* と動詞の左右に引数を取る両項型がある。

次の Fork と組み合わせれば相当複雑な構文も関数のみで表現できる。

A.1　Bond Compose　&

&　は数字と動詞を連結するときは　*bond*, 動詞と動詞を連結するときは *compose* という。

Bond(&) の機能は数字を連結することである。(Atop(@) はエラー)。 `0&{`　`1&{`

```
単項        両項
 u           u
 |      ↗        ↖
 v      v        v
 |      |        |
 y      x        y
```

$(\sim p) \wedge (\sim q)$

`x v&u y is (v x) u (v y)`

A.1.1　Atop

単項は&と同一でありどちらを用いても良い。

```
単項        両項
 u           u
 |           |
 v           v
 |      ↗        ↖
 y      x        y
```

```
   3+:@- 7
_8
```

付録 B　フックとフォーク

2 の動詞の組み合わせはフック、3 がフォークである。

動詞を連ねると *Train* になる。コンテナの貨物列車である。J は後ろから動詞を 3 組ずつ取って

いき、最後が 2 になればその部分はフックになる。

B.1　**フック**　Hook

洋服掛けのフックである。右に先に一つの作用をしてから次にかかる。フックは複雑なので余り用いない。

B.1.1　Hook

```
     単項             両項

       g                g
  ↗         ↖    ↗         ↖
  y          h    x          h
             |               |
             y               y
```

`-(+/ % #)`　`-"1(+/ % #)`　NB. 残差を計算する

B.2　Fork

`mean=:+/#%` のように引数を用いない動詞の定義ができる。

```
     単項                      両項

       g                         g
  ↗         ↖             ↗          ↖
  f          h        f                  h
  |          |     ↗      ↖         ↗      ↖
  y          y    x         y       x        y
```

```
   (+/%#) >:i.10
5.5
```

B.3　Capped fork

`[: u v` は Hook や Fork を掛けないで、通常の後ろから前の定義通りに実行したいときに用いる。

単項		両項	
g		g	
$\mid$		$\mid$	
h		h	
$\mid$	$\nearrow$		$\nwarrow$
y	x		y

付録 C　Reference

Milan Ondrus [Array processing with J] J wiki articles
K.E.Iverson [Calculous] http://jsoftware.com/Jwiki